
Drawing Commands
Commands:

Pt-On(



Draws point

Pt-Off(



Erases point

Pt-Change(


Changes whether a point is turned on or off

Pxl-On(


Turns a given pixel on

Pxl-Off(


Turns a given pixel off

Pxl-Change(


Changes whether a pixel is turned on or off

Pxl-Test(


Tests whether a given pixel is turned on or off

Horizontal


Draws a horizontal line

Vertical


Draws a vertical line

Line(



Draws a line

Circle(



Draws a circle

Using the Commands:

The point commands draw or erase a point on the graph screen.  Here are their syntaxes:
Pt-On(x,y,[type]
Pt-Off(x,y,[type]
where x is the x-coordinate and y is the y-coordinate of the point you want displayed and type is the type of point (1 is default and is a single pixel; 2 is a box; 3 is a cross).
The Pt-Change( command only takes two arguments and only changes a single pixel (we’ll see how this works a bit later).  Here’s the syntax for Pt-Change(:

Pt-Change(x,y

where x is the x-coordinate and y is the y-coordinate of the point you want changed.

The pixel commands all take two arguments and have the same syntax because they only affect a single pixel:
Pxl-On(row,column

Pxl-Off(row,column
Pxl-Change(row,column
Pxl-Test(row,column
where row is the number of pixels from the top of the screen and is an integer from 0 to 62 and column is the number of pixels from the left side of the screen and is an integer from 0 to 94.

To draw a horizontal line that goes all the way across the screen, use Horizontal:

Horizontal y
where y is the y-value where you want the line drawn.

To draw a vertical line that goes all the way from the top to the bottom of the screen, use the Vertical command:

Vertical x

where x is the x-value where you want the line drawn.
Line( can be used to draw or erase any type of line: vertical, horizontal, or diagonal.  Here is its syntax:

Line(x1,y1,x2,y2,[status]

where x1 and y1 are one endpoint of the line and x2 and y2 are the other endpoint.  The [status] can be 1 or 0, indicating on or off (default is 1).
To draw circles, use Circle(.  Here’s its syntax:

Circle(x,y,radius

where x and y are the coordinates of the center and radius is the radius.

Before using drawing commands, you may want to set the window to a specific range so that everything will always show up how you want it.  You can use ZStandard to set the Xmin and Ymin to -10 and the Xmax and Ymax to 10.  However, many programmers prefer to set the Xmin to 0, the Xmax to 94, the Ymin to -62, and the Ymax to 0.  This allows you to use the pixel and point commands together more easily.
Sample code:




Output:
[image: image1.png]


:ClrDraw:FnOff:AxesOff

:0→Xmin:94→Xmax

:-62→Ymin:0→Ymax

:Pt-On(47,-31
:Pause

:Pt-Off(47,-31

:Pause

:Pt-On(47,-31,2

:Pause

[image: image2.png]


:Pt-Off(47,-31,2

:Pause

:Pt-On(47,-31,3

:Pause

:Pt-Off(47,-31,3

:Pause

:Pt-Change(47,-31

:Pause

:Pt-Change(47,-31

:Pause

Explanation:

The first three lines of code set up the graph screen.  Notice that ClrDraw, FnOff, and AxesOff are all on the same line.  You can have as many commands as you want on the same line as long as they are separated by a colon.  The second and third lines of code set the window range so that our drawing commands will always output what we want, where we want it.  The first Pt-On( command outputs a single pixel in the middle of the screen.  Notice that there are only two arguments.  You could have Pt-On(47,-31,1, but the third argument is optional and default is 1 anyway, so we’ll save a couple bytes of memory by just leaving the third argument out.  After displaying our first point, the program pauses, then turns off the point.  The next Point-On( command displays a small square instead of just a single pixel because the type is set to 2 instead of 1.  The third Pt-On( command displays a small cross because the type is set to 3.  The first Pt-Change( command turns a pixel on because there is nothing there, but the second Pt-Change( command turns the pixel off.
Sample code:




Output:
[image: image3.png]


:ClrDraw:FnOff:AxesOff

:Pxl-On(47,31
:Pxl-Change(5,5

:Pause

:Pxl-Change(47,31

:Pxl-Off(5,5

:Pause

Explanation:
First, the screen is set up how we want it.  You don’t have to set the window range if you are only using pixel commands because each pixel is assigned a specific place on the screen and doesn’t rely on graphing coordinates to find its place like when using point or line commands.  In the program, two pixels are turned on: one near the bottom of the screen and one in the upper left of the screen.  The program pauses, then turns the pixels off.

Sample code:




Output:
[image: image4.png]


:ClrDraw:FnOff:AxesOff

:0→Xmin:94→Xmax

:-62→Ymin:0→Ymax

:Horizontal -31

:Vertical 47

:Line(0,0,46,-30

:Circle(70,-15,12
:Pause

:Line(0,0,46,-30,0

:Pause

Explanation:
This program draws a vertical line, a horizontal line, a diagonal line, and then a circle, then the program pauses.  Next, the diagonal line is erased by adding a 0 as a fifth argument of the Line( command.  Notice that the circle takes a really long time to draw, so don’t use it unless you have to until we learn about using pictures in a later tutorial.

Sample code:
:ClrDraw:FnOff:AxesOff
:0→Xmin:94→Xmax

:-62→Ymin:0→Ymax
:Lbl 1

:Pxl-Change(31,47



Output:
[image: image5.png]


:Pxl-Test(31,47→A
:If A=1

:Text(0,0,“THE PIXEL IS ON

:If A=0

:Text(0,0,“THE PIXEL IS OFF

:Pause

:Line(0,-1,56,-1,0

:Line(0,-2,56,-2,0

:Line(0,-3,56,-3,0

:Line(0,-4,56,-4,0

:Line(0,-5,56,-5,0

:Goto 1

Explanation:
This program introduces you to the infamous Pxl-Test( command, second in infamy only to the even more infamous getKey command.  First, the screen is set up how we want it.  Next, a pixel is turned on or off at (31,47) using Pxl-Change.  Then Pxl-Test( tests whether there is a pixel turned on at (31,47).  If there is a pixel turned on there, Pxl-Test( returns a 1, which is stored into variable A.  If there is no pixel turned on there, Pxl-Test( returns a 0, which is stored into variable A.  “If” statements test whether the value of A is 1 or 0 and then text is displayed to tell you whether the pixel at (31,47) is on or off.  What are all these line commands after the Pause?  Well, notice that they each have a fifth argument, which is 0.  These Line( commands simply allow us to erase the text without having to use a ClrDraw, thus preserving everything else on the screen.  This program loops using Lbl and Goto, so remember that you can break program execution by pressing [ON].
Conclusion:
Drawing commands are useful for creating a graphical game, or you can use them simply to enhance a text-based game with pictures.  Remember to set the window ranges before using these commands so that the calculator will always draw exactly what you want it to.  We didn’t go much into Pxl-Test( yet, but it really can’t be too useful until you know how to use loops and getKey, which we’ll cover in the next two tutorials.  Next, go to “Loops”.

