
Appendix D: Optimization
Once you’ve written your program, it’s time to optimize it.  In BASIC, we aren’t really concerned with program readability, but instead with speed and memory reduction.  In the first section of this appendix, we will be cutting out everything that is unnecessary to save every precious byte of memory we can.  In the second half, we will have some tips to maximize the speed of your program.
Saving Memory
Cut off ending quotes, parentheses, etc.

In TI BASIC, ending quotes, parentheses, brackets, and braces are unnecessary when at the end of a line.  Also, Store (→) acts as closing quotes, parentheses, brackets, and/or braces.
Instead of:

:Text(0,0,“HELLO WORLD!”)

Do:

:Text(0,0,“HELLO WORLD!

Instead of:

:{3,3}→dim([A])

Do:

:{3,3→dim([A]

Effectively arranging expressions (If statements and loops)

When using If statements and loops, think about how you can be efficient in writing the expression that is being tested.  Sometimes merely rearranging the expression can allow you to cut off ending quotes/parentheses/etc.  Never do something like “:If Aø0”; do “:If A” instead because the calculator interprets anything besides 0 as true.
Instead of:

:Repeat [A](A,B)=1

:commands
:commands

:End

Do:

:Repeat 1=[A](A,B
:commands
:commands

:End

Instead of:

:If Aø0

Do:

:If A

Instead of:

:While A=0

:commands
:commands

:End

Do:

:Repeat A

:commands
:commands

:End

Increasing Speed
Minimize use of labels

In most programming languages, the use of Goto and labels is considered outdated.  When searching for a label, the calculator starts at the top of the program and goes down line by line, so if you have a large program with labels at the bottom, the speed of your program can be greatly reduced.  If you really think that you do need to use labels (I am still guilty of this at times), place the most frequently accessed labels near the top of your program.  In order to completely eliminate the use of Goto and labels, you could do something like encasing most of your program in a While loop, then using If/Then/Else/End statements to perform a series of commands if a certain condition is met.  Another thing that people sometimes do is use Goto and Lbl for looping.  Don’t do it!  If you need to create loops, use a looping command like While, Repeat, or For(.
Instead of:

:Lbl 1

:commands

:commands

:Goto 1

Do:

:While 1

:commands

:commands

:End

Quickly evaluating keypress
The following tactic is especially good to use when you are using the arrow keys to move a sprite around the screen.  Instead of checking to see if a specific key was pressed and then storing new coordinates to your variables, use a mathematical expression so that you can just store coordinates to your variables.  (This was addressed somewhat in “The Infamous getKey Command”.)  Not only will this increase speed, it will also save some memory.

Instead of:

:While 1

:getKey→Z

:If Z=24

:A-1→A

:If Z=26

:A+1→A

:If Z=34

:B-1→B

:If Z=25

:B+1→B

:End

Do:

:While 1

:getKey→Z

:A-(Z=24)+(Z=26→A
:B-(Z=34)+(Z=25→B

:End

