Game: Snake

Introduction:

Ah, I’ve barely touched my calculator for the past two weeks because I’ve been learning to program for the Nintendo DS.  So… we’re ready to make another game.  It is by no means complicated; the only real concepts we are adding since our last game are the use of lists to store a highscore and (of course) custom menus.  We’ll be making Snake, a common game that you may have played before.  In this version, the snake is constantly getting longer and you must avoid yourself, the walls, and any pixels scattered on the screen.  If you really don’t feel like entering all the code yourself, just send SNAKE.8xp (included in the calctutorials folder) to your calculator and execute the program, but make sure that you go through all the code and read the explanation to make sure you understand everything that is going on.  Like with our last game, comments will be in blue.  One last thing: it’s been quite awhile since I actually programmed this, so it’s not fully optimized.  I’ll probably optimize it more later.  Here we go…
::"SNAKE  v2.1/ MDR  Falcon      

For use with shells
:ClrHome:ClrDraw:AxesOff:FnOff
:Text(0,38,"SNAKE



Title screen
:Text(22,21,"Programmed  by
:Text(29,30,"MDR  Falcon
:Text(55,41,"v2.1
:Pause :ClrDraw



Pauses the title screen
:0→Xmin:94→Xmax



Sets up a nice window for gameplay
:‾62→Ymin:0→YMax
:1→dim(∟SNK



Defines highscore list
:Lbl 1





Uh-oh… a While loop would be better
:ClrDraw
:Text(0,0,"Choose  difficulty:


Here’s our custom menu
:Text(14,14,"Easy
:Text(26,14,"Medium
:Text(38,14,"Hard
:Text(50,14,"Very  hard
:‾17→A




Initial Y-coordinate of arrow
:0→B





0 store to getKey variable
:1→C





Arrow will be turned on when loop begins
:While B≠105




Loops until [ENTER] is pressed
:getKey→B
:If B≠0:0→C




If key pressed, arrow will be turned off
:Line(5,A,11,A,C



Draw or erase arrow
:Line(10,A+1,10,A-1,C
:Line(9,A+2,9,A-2,C
:1→C





Will turn on the arrow
:A-12(B=34)(A>‾53)+12(B=25)(A<‾17→A
Moves arrow up or down if key is pressed
:End





Ends menu loop
:A→E:(‾E-5)/12→E



Harder difficulties will get more points
:(‾A-17)(25/6→A



Sets number of pixels to be drawn
:ClrDraw
:Horizontal(0:Horizontal(‾62


Draws boundaries
:Vertical 0:Vertical 94
:For(B,1,A




Draw a random pixel A times
:Pxl-On(randInt(1,61),randInt(1,93

Actually draw the pixel
:End





End pixel-drawing loop
:31→A:47→B




Starting coordinates of the snake
:26→C





Snake will begin game going right
:0→D





0 store to score variable
:Repeat pxl-Test(A,B



Main gameplay loop begins here
:Pxl-On(A,B




Draw a new pixel of the snake
:getKey
:If Ans
:Ans→C




Store getKey to C
:A+(C=34)-(C=25→A


Changes direction depending on arrow press
:B+(C=26)-(C=24→B



Changes direction depending on arrow press
:D+E→D




Add to score
:End





End of main gameplay loop
:If D>∟SNK(1:D→∟SNK(1

If highscore, store it to a list
:ClrHome
:Output(3,4,"GAME OVER
:Output(7,1,"Score:
:Output(7,7,D
:Output(8,1,"High:
:Output(8,6,∟SNK(1



Output highscore
:Pause :ClrHome:ClrDraw
:Text(0,0,"PLAY  AGAIN?


Another custom menu…
:Text(20,14,"Yes
:Text(36,14,"No
:‾23→A:0→B:1→C
:While B≠105
:getKey→B
:If B≠0:0→C
:Line(5,A,11,A,C
:Line(10,A+1,10,A-1,C
:Line(9,A+2,9,A-2,C
:1→C
:A-16(B=34)(A>‾39)+16(B=25)(A<‾23→A
:End
:If A=‾23:Goto 1:ClrDraw:Stop

If “Yes” was selected, jump to label 1
Explanation:
Fist we have our nice title screen and then the window is set for gameplay.  One is stored to the dimension of list SNK because if we didn’t do this, we risk getting an “Undefined” error later.

Next we have label 1, which is where the program will jump back to at the end of the game if the user wants to play another game.  A custom menu allows the user to select between four different difficulties.  The harder the difficulty, the more pixels there will be to avoid.  The initial Y-coordinate of the arrow is set to -17, then a loop begins to wait until the user makes a choice.  In the loop, the arrow is drawn if a key is not pressed and erased if a key is pressed.  Pressing the up and down arrow keys moves the arrow up or down by 12 pixels.  When [ENTER] is pressed, indicating that a selection has been made, the loop ends and the program moves on.
The next two lines make it so that harder difficulties will receive more points and more pixels to avoid will be drawn for those difficulties.  Our boundaries are drawn around the screen and a For( loop draws a number of random pixels based on the difficulty.  Next, the starting coordinates of the snake are set to 31 pixels from the top of the screen and 47 pixels from the left side of the screen.  Variable C will serve to store the direction that the snake is currently traveling in.  Here, it is set to 26 because we want it to move to the right in the beginning (26 is the getKey value of the right arrow key).  Variable D will keep track of the score, so it is set to 0.
Next, we have the main gameplay loop (starting with a Repeat( command), which is rather short.  The loop will essentially run until the snake runs into another pixel.  Within the loop, a getKey value is stored to variable C.  “getKey: If Ans: Ans→C” basically means, “If a key was pressed, stored the returned value to C.”  We don’t want 0 stored to our directional variable because it would cause the snake to stop moving.  Next, two mathematical expressions store new coordinates to the snake’s front based on the direction in variable C, making the snake longer.  The user’s score is then increased based on the difficulty level.

After the main gameplay loop, the user’s score is stored to the highscore list if it is greater than the score already stored there.  A “GAME OVER” message and the score are displayed.  Then we have a custom menu that asks the user if they want to play again.  (I won’t analyze the menu here because it is basically the same as the first custom menu in the program.  If you need a review on custom menus, see the last tutorial.)  If “Yes” is selected, the program jumps to label 1.  Otherwise, the program ends.
Conclusion:

That should really have been a piece of cake to understand if you understood all the tutorials up to this point.  The next tutorial will be a specialized one on animation.  Then we will have (drumroll, please) an introduction to z80 Assembly language, which will be difficult but rewarding.  Then (another drumroll, please) you will be done learning from me, unless I come up with some more ideas for tutorials.

