
The Infamous getKey Command
Commands:

getKey



Returns numerical code of the current keypress

Using the Commands:

At last we’ve come to the infamous getKey command!  What does this wondrous command do?  It can tell you if the user pressed a key and which key that was.  This will open up a world of opportunity and will finally allow you to make (gasp!) real, non text-based games (not that there’s anything wrong with text-based games, of course).  So, how do you use this command?  Usually it is stored into a variable like this:
:getKey→A

Then, an If statement can check what is stored in variable A or you can use the value of A in a mathematical expression.  Each key has a specific numerical code assigned to it, as shown in the diagram to the right.
Before we get to the sample programs, we need to learn about how relational operators work so that we can make our programming more efficient.  Take for example the following expression: (A<7).  When the calculator evaluates this expression, it evaluates it as either true or false.  In the world of computers and calculators, 1 represents true and 0 represents false.  So, in this case, if A was less than seven, the calculator would return a 1 for true.  If A was not less than seven, however, the calculator would return a 0 for false.  You’ll see how we can use this to our advantage when working with getKey soon.
Sample code:




Output:

:ClrHome

:0→A
:While 1
:getKey→A

:If Aø0

:Disp A

:If A=45

:Stop

:End

Explanation:

When you press a key, the code of that key is displayed on the homescreen.  A While loop is used to create an infinite loop.  Inside the loop, the value that getKey returns is stored to variable A.  If A is not equal to 0 (indicating that a key was pressed), the value that getKey returned is displayed.  If [CLEAR] is pressed (which has a keycode of 45), the program stops.
Sample code:




Output:

:ClrDraw:FnOff:AxesOff

:0→Xmin:94→Xmax

:-62→Ymin:0→Ymax

:47→A:-31→B

:While 1

:Pt-On(A,B,3

:getKey→Z

:If Zø0:Pt-Off(A,B,3

:If Z=45:Stop

:If Z=24:A-1→A

:If Z=26:A+1→A

:If Z=34:B-1→B

:If Z=25:B+1→B

:End

Explanation:
This program allows you to move a small cross around the screen.  First, the program sets up the graphscreen how we want it.  Next, it stores the initial coordinates of our cross into A and B.  Inside the While loop, a cross point is displayed at (A,B).  If a key is pressed (Z is not equal to 0), the point at (A,B) is turned off.  If [CLEAR] is pressed, the program stops.  The next four lines check to see if one of the arrow keys was pressed.  The appropriate coordinate variable is then incremented or decremented according to which key was pressed, and then the loop starts over again.

Sample code:




Output:

:ClrDraw:FnOff:AxesOff

:0→Xmin:94→Xmax

:-62→Ymin:0→Ymax

:47→A:-31→B

:While 1

:Pt-On(A,B,3

:getKey→Z
:If Zø0:Pt-Off(A,B,3

:If Z=45:Stop
:A-(Z=24)+(Z=26)→A

:B-(Z=34)+(Z=25)→B

:End

Explanation:
That program did the same thing as the second program, but in a more efficient way.  The expressions in the parentheses return a 1 if true and a 0 if false.  Assume that the left arrow key was pressed, so getKey would return a 24 and that would be stored to Z.  So then :A-(Z=24)+(Z=26)→A would be equivalent to A-(1)+(0)→A, which is like A-1→A and :B-(Z=34)+(Z=25)→B would be equivalent to B-(0)+(0)→B, which does nothing to change the value of B.  With the previous two programs, it is possible to move the cross off the screen.  How could we fix this?  Use more relational operators, of course!

Sample code:




Output:

:ClrDraw:FnOff:AxesOff

:0→Xmin:94→Xmax

:-62→Ymin:0→Ymax

:47→A:-31→B

:While 1

:Pt-On(A,B,3

:getKey→Z
:If Zø0:Pt-Off(A,B,3

:If Z=45:Stop
:A-(A>0)(Z=24)+(A<94)(Z=26)→A

:B-(B>-62)(Z=34)+(B<0)(Z=25)→B

:End

Explanation:

Try it!  The cross won’t be able to move off the screen.  When it gets to the edge, it stops.  Assume again that the left arrow key was pressed and the cross was at (7,-42).  That means that A>0 and A<94, so :A-(A>0)(Z=24)+(A<94)(Z=26)→A is like                      A-(1)(1)+(1)(0)→A, which is equivalent to A-1→A.  Now assume that the cross is at (0,-42).  That means that A is not greater than 0 and A is less than 94.  That means that      :A-(A>0)(Z=24)+(A<94)(Z=26)→A is like A-(0)(1)+(1)(0), which results in the value of A not being changed.  What if you wanted to use lines to determine the boundaries of where the cross can move?  Looks like it’s time for more sample code…

Sample code:




Output:
:ClrDraw:FnOff:AxesOff

:0→Xmin:94→Xmax

:-62→Ymin:0→Ymax

:Vertical 0:Vertical 94

:Horizontal 0:Horizontal -62

:Line(20,-42,74,-42

:47→A:-31→B
:A→C:-B→D

:While 1

:Pt-On(A,B

:getKey→Z

:If Zø0:Pt-Off(A,B

:If Z=45:Stop

:C-(Z=24)+(Z=26)→C
:D+(Z=34)-(Z=25)→D

:Pxl-Test(D,C→E

:If E=0:Then

:C→A:-D→B

:End

:If E=1:Then
:A→C:-B→D

:End

:End

Explanation:

This program is a bit more complex than most of the others we’ve done so far.  First, the graph screen is set up and some lines are drawn on it.  The initial coordinates of our point are set and the loop begins.  Notice that we are only using a single pixel for our point instead of a cross.  We could use a cross, but the method of testing for a turned-on pixel would be more complex.  Also notice that we are using variables C and D in addition to A and B.  The reason we are doing this is to “scout ahead” and still be able to retain the coordinates of our point should the “scouted” pixel be turned on.  After the two lines that change C and D based on what key is pressed, a pixel test is conducted in order to see if the pixel we want our point to move to is turned on.  If the pixel is turned off, new coordinates are set for A and B, in essence moving our point.  If the pixel is turned on, the coordinates of C and D are reset as they were at the beginning of the loop and our point is not allowed to move.  Anyway, try moving the point through the lines.  It won’t work.  The point will just stop when it’s next to the line.
Conclusion:

I’m sure (or at least I hope) that this tutorial was a little bit more interesting than the last.  Here, we learned about using keypress as user input, how loops can be used for things that are actually useful (not to say that counting isn’t useful), and a little bit more on how to use Pxl-Test(.  Now that we understand all these things, we are almost ready to make a fully graphical game, but we’ll be learning how to store and recall pictures before we do that.  Loops and getKey can also be used to create your own custom menus, which we’ll get to in due time.  Go to “Pictures”.

